

Norms of Vectors and Matrices

```
> restart;
```

We begin with the **LinearAlgebra** package.

```
> with(LinearAlgebra):
```

We input a vector.

```
> x:=Vector(4,[5,-3,-6,2]);
```

$$x := \begin{bmatrix} 5 \\ -3 \\ -6 \\ 2 \end{bmatrix}$$

The default [Norm](#) is the sup or l_∞ norm.

```
> Norm(x);
```

6

```
> Norm(x,infinity);
```

6

We do the following to get the l_2 norm.

```
> Norm(x,2);
```

$\sqrt{74}$

```
> Norm(x,Euclidean);
```

$\sqrt{74}$

We want to find the [limit](#) of a sequence of vectors. This is based on a theorem in the notes and text.

```
> v:=Vector(4,[3+5/k,5,2/k^2,exp(-k)*cos(k)]);
```

$$v := \begin{bmatrix} 3 + \frac{5}{k} \\ 5 \\ \frac{2}{k^2} \\ e^{-k} \cos(k) \end{bmatrix}$$

```
> limvec:=Vector(4):  
for i from 1 to 4 do  
limvec[i]:=limit(v[i],k=infinity);  
od;  
vectorlimit:=limvec;
```

$limvec_1 := 3$

$limvec_2 := 5$

$limvec_3 := 0$

$limvec_4 := 0$

$$\text{vectorlimit} := \begin{bmatrix} 3 \\ 5 \\ 0 \\ 0 \end{bmatrix}$$

An easier way to find a vector limit is to use the [VectorLimit](#) command in the [NumericalAnalysis](#) package.

```
> with(Student[NumericalAnalysis]);
[AbsoluteError, AdamsBashforth, AdamsBashforthMoulton, AdamsMoulton, AdaptiveQuadrature,
AddPoint, ApproximateExactUpperBound, ApproximateValue, BackSubstitution, BasisFunctions,
Bisection, CubicSpline, DataPoints, Distance, DividedDifferenceTable, Draw, Euler, EulerTutor,
ExactValue, FalsePosition, FixedPointIteration, ForwardSubstitution, Function,
InitialValueProblem, InitialValueProblemTutor, Interpolant, InterpolantRemainderTerm,
IsConvergent, IsMatrixShape, IterativeApproximate, IterativeFormula, IterativeFormulaTutor,
LeadingPrincipalSubmatrix, LinearSolve, LinearSystem, MatrixConvergence,
MatrixDecomposition, MatrixDecompositionTutor, ModifiedNewton, NevilleTable, Newton,
NumberOfSignificantDigits, PolynomialInterpolation, Quadrature, RateOfConvergence,
RelativeError, RemainderTerm, Roots, RungeKutta, Secant, SpectralRadius, Steffensen, Taylor,
TaylorPolynomial, UpperBoundOfRemainderTerm, VectorLimit]
```

```
> vectorlimit1:=VectorLimit(v,k);
```

$$\text{vectorlimit1} := \begin{bmatrix} 3 \\ 5 \\ 0 \\ 0 \end{bmatrix}$$

We return to the **LinearAlgebra** package to look at matrix norms.

```
> A:=Matrix(4,4,[[1,1,3,4],[1,-1,1,1],[3,-1,2,-3],[2,3,-1,4]]);
```

$$A := \begin{bmatrix} 1 & 1 & 3 & 4 \\ 1 & -1 & 1 & 1 \\ 3 & -1 & 2 & -3 \\ 2 & 3 & -1 & 4 \end{bmatrix}$$

The default norm is again the sup or l_∞ norm.

```
> Norm(A);
```

10

```
> Norm(A,infinity);
```

10

Calculating the sup norm from the matrix entries. Note the use of [Row](#).

```
> for i from 1 to 4 do
R[i]:=Row(A,i);
r[i]:=add(abs(R[i][j]),j=1..4);
```

```

od;
supnorm:=max(r[1],r[2],r[3],r[4]);
      R1 := [ 1 1 3 4 ]
          r1 := 9
      R2 := [ 1 -1 1 1 ]
          r2 := 4
      R3 := [ 3 -1 2 -3 ]
          r3 := 9
      R4 := [ 2 3 -1 4 ]
          r4 := 10
      supnorm := 10

```

We find the l_2 norm.

```

> Norm(A,2);
       $\sqrt{\text{RootOf}(12321 + \_Z^4 - 84 \_Z^3 + 1918 \_Z^2 - 12549 \_Z, \text{index} = 4)}$ 
> evalf(%);
      7.16073839926786

```

```

> Norm(A,Euclidean);
       $\sqrt{\text{RootOf}(12321 + \_Z^4 - 84 \_Z^3 + 1918 \_Z^2 - 12549 \_Z, \text{index} = 4)}$ 
> evalf(%);
      7.16073839926786

```

We return to the **NumericalAnalysis** package to see if the matrix A is **convergent**.

```

> MatrixConvergence(A);
      false

```

We check by its spectral radius.

```

> SpectralRadius(A);
       $\frac{3}{2} + \frac{1}{2} \sqrt{59 + 2\sqrt{181}}$ 

```

This is clearly greater than 1.